tetradTest.nb

For the given tetrad, this notebook computes the left-hand-side of the equations of motion within the (a) non-Lorentz-covariant and the (b) Lorentz-covariant form of F(T) gravity theory.

The notebook can be considered as supplementary material to arXiv:1609:07465 by A. DeBenedictis and S. Ilijic.

```
In[1]:= Date[]

Out[1]= {2016, 11, 24, 0, 5, 57.242177}
```

1. Coordinates and the tetrad

The four specetime coordinates and the number of dimensions:

```
In[2]:= cu = {t, r, ϑ, φ};
        ndim = Length[cu];
```

Tetrad (first index is the tetrad-up-index, second index is the spacetime-down-index, therefore variable name ends with Ud):

```
In[4]:= hUd = ( A[r]   0     0    0
                 0     B[r]  0    0
                 0     0     r    0
                 0     0     0    r Sin[ϑ] );
```

The tetrad with first (tetrad) index up and second (spacetime) index down:

```
In[5]:= hDu = Transpose@Inverse[hUd];
```

The transformation rule that enforces the flat spacetime limit:

```
In[6]:= flatRule = {A → Function[r, 1], B → Function[r, 1]}

Out[6]= {A → Function[r, 1], B → Function[r, 1]}
```

The determinant of the tetrad matrix:

```
In[7]:= dethUd = Det[hUd] // Simplify

Out[7]= r^2 A[r] B[r] Sin[ϑ]
```

2. Metric compatibility

The tetrad gives raise to the following metric:

```
In[8]:= mink = -IdentityMatrix[ndim]; mink[[1, 1]] *= -1;
        gdd = Transpose[hUd].mink.hUd // Simplify;
        gdd // MatrixForm

Out[10]//MatrixForm=
        ( A[r]^2   0         0      0
          0        -B[r]^2   0      0
          0        0         -r^2   0
          0        0         0      -r^2 Sin[ϑ]^2 )
```

The **flatRule** transforms the metric tensor into its flat-space limit:

In[11]:=
```
gdd /. flatRule // MatrixForm
```

Out[11]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -r^2 & 0 \\ 0 & 0 & 0 & -r^2\,\mathrm{Sin}[\vartheta]^2 \end{pmatrix}$$

The up-up index meric tensor:

In[12]:=
```
guu = Inverse[gdd];
guu // MatrixForm
```

Out[13]//MatrixForm=

$$\begin{pmatrix} \frac{1}{A[r]^2} & 0 & 0 & 0 \\ 0 & -\frac{1}{B[r]^2} & 0 & 0 \\ 0 & 0 & -\frac{1}{r^2} & 0 \\ 0 & 0 & 0 & -\frac{\mathrm{Csc}[\vartheta]^2}{r^2} \end{pmatrix}$$

3. Torsion tensor first part ($T^{(0)}$, assuming spin connection = 0)

Torsion tensor:

In[14]:=
```
tmpUdd = Transpose[D[hUd, #] & /@ cu];
torsion0Udd = tmpUdd - Transpose[tmpUdd, {1, 3, 2}] // Simplify;
torsion0udd = Transpose[hDu].torsion0Udd // Simplify;
MatrixForm /@ %
```

Out[17]=

$$\left\{ \begin{pmatrix} 0 & -\frac{A'[r]}{A[r]} & 0 & 0 \\ \frac{A'[r]}{A[r]} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{r} & 0 \\ 0 & -\frac{1}{r} & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{r} \\ 0 & 0 & 0 & \mathrm{Cot}[\vartheta] \\ 0 & -\frac{1}{r} & -\mathrm{Cot}[\vartheta] & 0 \end{pmatrix} \right\}$$

4. Spin connection (ω, with tracking factor" σ)

Calculation of the spin connection as proposed by Krssak and Saridakis (arXiv:1510.08432):

In[18]:=
```
omgUDd = Array[omg, {4, 4, 4}];
omgUdd = Transpose[Transpose[hUd].Transpose[omgUDd]] /. flatRule;
```

In[20]:=
```
torUdd = torsion0Udd + (Transpose[omgUdd, {1, 3, 2}] - omgUdd) // Simplify;
```

In[21]:=
```
omgEqns1 = 0 == # & /@ Flatten[torUdd /. flatRule];
```

In[22]:=
```
omgEqns2 = 0 == # & /@ Flatten[mink.omgUDd + Transpose[mink.omgUDd]];
```

In[23]:=
```
omgsol = Solve[Join[omgEqns1, omgEqns2], Flatten[omgUDd]];
```

Spin connection omega multiplied by the "tracking-parameter" sigma (first index is tetrad-up, second is tetrad-down, third is spacetime-down):

In[24]:=
```
omegaUDd = σ omgUDd /. omgsol[[1]] // Simplify
```

Out[24]=
```
{{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
 {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, -σ, 0}, {0, 0, 0, -σ Sin[ϑ]}},
 {{0, 0, 0, 0}, {0, 0, σ, 0}, {0, 0, 0, 0}, {0, 0, 0, -σ Cos[ϑ]}},
 {{0, 0, 0, 0}, {0, 0, 0, σ Sin[ϑ]}, {0, 0, 0, σ Cos[ϑ]}, {0, 0, 0, 0}}}
```

Testing that the four matrices are antisymmetric:

In[25]:=
```
MatrixForm /@ Transpose[mink.omegaUDd, {2, 3, 1}]
```

Out[25]=
$$\left\{\begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&\sigma&0\\0&-\sigma&0&0\\0&0&0&0\end{pmatrix}, \begin{pmatrix}0&0&0&0\\0&0&0&\sigma\,\mathrm{Sin}[\vartheta]\\0&0&0&\sigma\,\mathrm{Cos}[\vartheta]\\0&-\sigma\,\mathrm{Sin}[\vartheta]&-\sigma\,\mathrm{Cos}[\vartheta]&0\end{pmatrix}\right\}$$

Spin connection (first two indices are tetrad-up, third is spacetime-down):

In[26]:=
```
omegaUUd = Transpose[mink.Transpose[omegaUDd]]
```

Out[26]=
```
{{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
 {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, σ, 0}, {0, 0, 0, σ Sin[ϑ]}},
 {{0, 0, 0, 0}, {0, 0, -σ, 0}, {0, 0, 0, 0}, {0, 0, 0, σ Cos[ϑ]}},
 {{0, 0, 0, 0}, {0, 0, 0, -σ Sin[ϑ]}, {0, 0, 0, -σ Cos[ϑ]}, {0, 0, 0, 0}}}
```

Some other variants for later convenience

In[27]:=
```
omegaUdd = Transpose[Transpose[hUd].Transpose[omegaUDd]];
tmpuud = Transpose[hDu].Transpose[Transpose[hDu].Transpose[omegaUUd]];
omegau = Table[Sum[tmpuud[[α, μ, α]], {α, 4}], {μ, 4}] // Simplify
```

Out[29]=
$$\left\{0, -\frac{2\sigma}{r\,B[r]}, -\frac{\sigma\,\mathrm{Cot}[\vartheta]}{r^2}, 0\right\}$$

5. Complete torsion tensor (*T*, now involves spin connection)

This is now the complete torsion tensor which involves the spin connection:

In[30]:=
```
torsionUdd = torsion0Udd + (Transpose[omegaUdd, {1, 3, 2}] - omegaUdd) // Simplify;
torsionudd = Transpose[hDu].torsionUdd // Simplify
```

Out[31]=
$$\left\{\left\{\left\{0, -\frac{A'[r]}{A[r]}, 0, 0\right\}, \left\{\frac{A'[r]}{A[r]}, 0, 0, 0\right\}, \{0, 0, 0, 0\}, \{0, 0, 0, 0\}\right\},\right.$$
$$\{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}, \{0, 0, 0, 0\}, \{0, 0, 0, 0\}\},$$
$$\left\{\{0, 0, 0, 0\}, \left\{0, 0, \frac{1-\sigma B[r]}{r}, 0\right\}, \left\{0, \frac{-1+\sigma B[r]}{r}, 0, 0\right\}, \{0, 0, 0, 0\}\right\}, \left\{\{0, 0, 0, 0\},\right.$$
$$\left.\left.\left\{0, 0, 0, \frac{1-\sigma B[r]}{r}\right\}, \{0, 0, 0, -(-1+\sigma)\,\mathrm{Cot}[\vartheta]\}, \left\{0, \frac{-1+\sigma B[r]}{r}, (-1+\sigma)\,\mathrm{Cot}[\vartheta], 0\right\}\right\}\right\}$$

In[32]:=
```
torsionddd = gdd.torsionudd // Simplify;
torsionuuu = Transpose[guu.Transpose[torsionudd.guu]] // Simplify;
```

In[34]:=
```
torsiond = Apply[Plus, #] & /@ Transpose[torsionudd, {2, 1, 2}] // Simplify
```

Out[34]=
$$\left\{0, \frac{2-2\sigma B[r]}{r} + \frac{A'[r]}{A[r]}, -(-1+\sigma)\,\mathrm{Cot}[\vartheta], 0\right\}$$

6. Contortion (*K*)

This is the contortion tensor:

In[35]:=
```
contortionddd =
  (Transpose[gdd.torsionudd, {1, 3, 2}] + Transpose[gdd.torsionudd, {2, 1, 3}] +
     Transpose[gdd.torsionudd, {3, 1, 2}]) / 2 // Simplify;
contortionudd = guu.contortionddd // Simplify;
MatrixForm /@ contortionudd
```

Out[37]=

$$\left\{\begin{pmatrix} 0 & 0 & 0 & 0 \\ -\frac{A'[r]}{A[r]} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} -\frac{A[r]\,A'[r]}{B[r]^2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{r-r\,\sigma\,B[r]}{B[r]^2} & 0 \\ 0 & 0 & 0 & -\frac{r\,(-1+\sigma\,B[r])\,\mathrm{Sin}[\vartheta]^2}{B[r]^2} \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{-1+\sigma\,B[r]}{r} & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -(-1+\sigma)\,\mathrm{Cos}[\vartheta]\,\mathrm{Sin}[\vartheta] \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{-1+\sigma\,B[r]}{r} \\ 0 & 0 & 0 & (-1+\sigma)\,\mathrm{Cot}[\vartheta] \\ 0 & 0 & 0 & 0 \end{pmatrix}\right\}$$

7. Modified torsion or "superpotential" (*S*)

This is the modified torsion tensor or the superpotential:

In[38]:=
```
modtorsionddd =
  (Transpose[contortionddd, {2, 3, 1}] + Transpose[Outer[Times, gdd, torsiond], {1, 2, 3}] -
    Transpose[Outer[Times, gdd, torsiond], {1, 3, 2}]) // Simplify;
MatrixForm /@
 %
```

Out[39]=

$$\left\{\begin{pmatrix} 0 & -\frac{2\,A[r]^2\,(-1+\sigma\,B[r])}{r} & -(-1+\sigma)\,A[r]^2\,\mathrm{Cot}[\vartheta] & 0 \\ \frac{2\,A[r]^2\,(-1+\sigma\,B[r])}{r} & 0 & 0 & 0 \\ (-1+\sigma)\,A[r]^2\,\mathrm{Cot}[\vartheta] & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & (-1+\sigma)\,B[r]^2\,\mathrm{Cot}[\vartheta] & 0 \\ 0 & -(-1+\sigma)\,B[r]^2\,\mathrm{Cot}[\vartheta] & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & r\left(1-\sigma\,B[r]+\frac{r\,A'[r]}{A[r]}\right) & 0 \\ 0 & r\left(-1+\sigma\,B[r]-\frac{r\,A'[r]}{A[r]}\right) & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{r\,\mathrm{Sin}[\vartheta]^2\,(A[r]-\sigma\,A[r]\,B[r]+r\,A'[r])}{A[r]} \\ 0 & 0 & 0 & 0 \\ 0 & -\frac{r\,\mathrm{Sin}[\vartheta]^2\,(A[r]-\sigma\,A[r]\,B[r]+r\,A'[r])}{A[r]} & 0 & 0 \end{pmatrix}\right\}$$

In[40]:=
```
modtorsionuuu = Transpose[guu.Transpose[guu.modtorsionddd.guu]] // Simplify;
modtorsionDuu = mink.hUd.modtorsionuuu // Simplify;
```

8. Torsion scalar (T)

This is the torsion scalar with the spin-connection-tracking-factor sigma:

In[42]:=
```
torsionscalar =
  Sum[torsionuuu[[a, b, c]] modtorsionddd[[a, b, c]], {a, ndim}, {b, ndim}, {c, ndim}] / 2 //
   Simplify
```

Out[42]= $-\frac{2\,(-1+\sigma\,B[r])\,(A[r]-\sigma\,A[r]\,B[r]+2\,r\,A'[r])}{r^2\,A[r]\,B[r]^2}$

Setting sigma=0 gives the torsion scalar used in the standard TEGR and in the non-covariant form of f(T)-gravity:

In[43]:=
```
torsionscalar /. {σ → 0}
```

Out[43]= $\frac{2\,(A[r]+2\,r\,A'[r])}{r^2\,A[r]\,B[r]^2}$

Setting sigma=1 gives the torsion scalar used in the covariant form of F(T)-gravity.

In[44]:=
```
torsionscalar /. {σ → 1}
```

Out[44]= $-\frac{2\,(-1+B[r])\,(A[r]-A[r]\,B[r]+2\,r\,A'[r])}{r^2\,A[r]\,B[r]^2}$

9. Equations of motion for f(T) gravity

These are the terms of the left-hand-side of the equation of motion (see 1609:07465 eq. (10)):

In[45]:=
```
foftEomTerm1aDu = foft2 modtorsionDuu.(D[torsionscalar, #] & /@ cu);
foftEomTerm1bDu = dethUd^-1 foft1
   Table[Sum[(D[dethUd modtorsionDuu, #] & /@ cu)[[ν, a, μ, ν]], {ν, 4}], {a, 4}, {μ, 4}];
foftEomTerm2Du = -foft1 Table[ Sum[(torsionUdd.Transpose[hDu])[[c, ν, a]]
      modtorsionDuu[[c, ν, μ]], {c, 4}, {ν, 4}], {a, 4}, {μ, 4}];
foftEomTerm3Du = foft0 hDu / 2;
foftEomTerm4Du = foft1
   Table[ Sum[omegaUDd[[c, a, ν]] modtorsionDuu[[c, ν, μ]], {c, 4}, {ν, 4}], {a, 4}, {μ, 4}];
```

The left-hand-side of the equations of motion:

In[50]:=
```
foftEomLhsDu =
  foftEomTerm1aDu + foftEomTerm1bDu + foftEomTerm2Du + foftEomTerm3Du + foftEomTerm4Du;
foftEomLhsdu = Transpose[hUd].foftEomLhsDu // Simplify;
```

In static spherical symmetry one expects diagonal energy-momentum tensor and therefore also the left-hand-side of the equations of motion (left plot of the matrix is the non-covariant form, right plot is the covariant form):

In[52]:=
```
GraphicsArray[{MatrixPlot[foftEomLhsdu /. {σ → 0} // Simplify],
  MatrixPlot[foftEomLhsdu /. {σ → 1} // Simplify]}]
```

Out[52]=



In[53]:= **foftEomLhsdu /. {σ → 0} // Simplify**

Out[53]=
```
{{1/(2 r^4 A[r]^2 B[r]^5)
   (16 foft2 r^2 B[r] A′[r]^2 + A[r]^2 (16 foft2 B[r] - 4 foft1 r^2 B[r]^3 + (2 foft1 r^2 + foft0 r^4) B[r]^5 +
        16 foft2 r B′[r] + 4 foft1 r^3 B[r]^2 B′[r]) -
     4 r A[r] (foft1 r^2 B[r]^3 A′[r] - 8 foft2 r A′[r] B′[r] - 4 foft2 B[r] (A′[r] - r A″[r]))),
  0, 0, 0}, {0, foft0/2 + foft1/r^2 - (2 foft1 (A[r] + 2 r A′[r]))/(r^2 A[r] B[r]^2),
  - 1/(r^5 A[r]^2 B[r]^3) 4 foft2 Cot[ϑ]
     (r^2 B[r] A′[r]^2 + A[r]^2 (B[r] + r B′[r]) + r A[r] (2 r A′[r] B′[r] + B[r] (A′[r] - r A″[r]))), 0},
 {0, 0, 1/(2 r^4 A[r]^3 B[r]^5) (8 foft2 r^3 B[r] A′[r]^3 + A[r]^3
      (8 foft2 B[r] - 2 foft1 r^2 B[r]^3 + foft0 r^4 B[r]^5 + 8 foft2 r B′[r] + 2 foft1 r^3 B[r]^2 B′[r]) +
     8 foft2 r^2 A[r] A′[r] (2 r A′[r] B′[r] + B[r] (2 A′[r] - r A″[r])) -
     2 r A[r]^2 (-12 foft2 r A′[r] B′[r] - foft1 r^3 B[r]^2 A′[r] B′[r] +
        foft1 r^2 B[r]^3 (3 A′[r] + r A″[r]) + B[r] (-8 foft2 A′[r] + 4 foft2 r A″[r]))), 0},
 {0, 0, 0, 1/(2 r^4 A[r]^3 B[r]^5) (8 foft2 r^3 B[r] A′[r]^3 + A[r]^3
      (8 foft2 B[r] - 2 foft1 r^2 B[r]^3 + foft0 r^4 B[r]^5 + 8 foft2 r B′[r] + 2 foft1 r^3 B[r]^2 B′[r]) +
     8 foft2 r^2 A[r] A′[r] (2 r A′[r] B′[r] + B[r] (2 A′[r] - r A″[r])) -
     2 r A[r]^2 (-12 foft2 r A′[r] B′[r] - foft1 r^3 B[r]^2 A′[r] B′[r] +
        foft1 r^2 B[r]^3 (3 A′[r] + r A″[r]) + B[r] (-8 foft2 A′[r] + 4 foft2 r A″[r])))}}
```

In[54]:= **foftEomLhsdu /. {σ → 1} // Simplify**

Out[54]=
```
{{1/(2 r^4 A[r]^2 B[r]^5) (16 foft2 r^2 (-1 + B[r])^2 B[r] A'[r]^2 +
    A[r]^2 ((48 foft2 - 4 foft1 r^2) B[r]^3 + 4 (-4 foft2 + foft1 r^2) B[r]^4 + foft0 r^4 B[r]^5 + 16 foft2 r
        B'[r] + 16 foft2 B[r] (1 - 2 r B'[r]) + 4 B[r]^2 (-12 foft2 + (4 foft2 r + foft1 r^3) B'[r])) +
    4 r A[r] (-1 + B[r]) (foft1 r^2 B[r]^3 A'[r] - 8 foft2 r A'[r] B'[r] +
       4 foft2 B[r]^2 (A'[r] - r A''[r]) + 4 foft2 B[r] (A'[r] (-1 + r B'[r]) + r A''[r]))), 0, 0, 0},
  {0, (A[r] (-4 foft1 + 4 foft1 B[r] + foft0 r^2 B[r]^2) + 4 foft1 r (-2 + B[r]) A'[r])/(2 r^2 A[r] B[r]^2),
   0,
   0}, {0, 0,
   1/(2 r^4 A[r]^3 B[r]^5)
   (-8 foft2 r^3 (-1 + B[r]) B[r] A'[r]^3 + A[r]^3 ((24 foft2 - 2 foft1 r^2) B[r]^3 +
        (-8 foft2 + 4 foft1 r^2) B[r]^4 + (-2 foft1 r^2 + foft0 r^4) B[r]^5 + 8 foft2 r B'[r] +
        8 foft2 B[r] (1 - 2 r B'[r]) + 2 B[r]^2 (-12 foft2 + (4 foft2 r + foft1 r^3) B'[r])) +
     8 foft2 r^2 A[r] A'[r] (B[r]^3 A'[r] + 2 r A'[r] B'[r] + B[r]^2 (-3 A'[r] + r A''[r]) -
        B[r] (A'[r] (-2 + r B'[r]) + r A''[r])) - 2 r A[r]^2 (-2 foft1 r^2 B[r]^4 A'[r] -
        12 foft2 r A'[r] B'[r] + 4 foft2 B[r] (A'[r] (-2 + 4 r B'[r]) + r A''[r]) -
        B[r]^2 (A'[r] (-16 foft2 + (4 foft2 r + foft1 r^3) B'[r]) + 8 foft2 r A''[r]) +
        B[r]^3 ((-8 foft2 + 3 foft1 r^2) A'[r] + r (4 foft2 + foft1 r^2) A''[r]))), 0},
  {0, 0, 0, 1/(2 r^4 A[r]^3 B[r]^5) (-8 foft2 r^3 (-1 + B[r]) B[r] A'[r]^3 +
     A[r]^3 ((24 foft2 - 2 foft1 r^2) B[r]^3 + (-8 foft2 + 4 foft1 r^2) B[r]^4 +
        (-2 foft1 r^2 + foft0 r^4) B[r]^5 + 8 foft2 r B'[r] + 8 foft2 B[r] (1 - 2 r B'[r]) +
        2 B[r]^2 (-12 foft2 + (4 foft2 r + foft1 r^3) B'[r])) +
     8 foft2 r^2 A[r] A'[r] (B[r]^3 A'[r] + 2 r A'[r] B'[r] + B[r]^2 (-3 A'[r] + r A''[r]) -
        B[r] (A'[r] (-2 + r B'[r]) + r A''[r])) - 2 r A[r]^2 (-2 foft1 r^2 B[r]^4 A'[r] -
        12 foft2 r A'[r] B'[r] + 4 foft2 B[r] (A'[r] (-2 + 4 r B'[r]) + r A''[r]) -
        B[r]^2 (A'[r] (-16 foft2 + (4 foft2 r + foft1 r^3) B'[r]) + 8 foft2 r A''[r]) +
        B[r]^3 ((-8 foft2 + 3 foft1 r^2) A'[r] + r (4 foft2 + foft1 r^2) A''[r]))))}}
```

The difference between the non-covariant and the covariant form of the left-hand-side of the equations of motion:

In[55]:=
```
(foftEomLhsdu /. {σ → 0}) - (foftEomLhsdu /. {σ → 1}) // Simplify
```

Out[55]=
```
{{1/(r^4 A[r]^2 B[r]^4)
   (-8 foft2 r^2 (-2 + B[r]) B[r] A′[r]^2 + A[r]^2 (-24 foft2 B[r]^2 + (8 foft2 - 2 foft1 r^2) B[r]^3 +
       foft1 r^2 B[r]^4 + 16 foft2 r B′[r] - 8 foft2 B[r] (-3 + r B′[r])) -
    2 r A[r] (foft1 r^2 B[r]^3 A′[r] - 12 foft2 r A′[r] B′[r] + 4 foft2 B[r]^2 (A′[r] - r A″[r]) +
       4 foft2 B[r] (A′[r] (-2 + r B′[r]) + 2 r A″[r]))), 0, 0, 0},
 {0, (foft1 (A[r] (-2 + B[r]) - 2 r A′[r]))/(r^2 A[r] B[r]), -1/(r^5 A[r]^2 B[r]^3) 4 foft2 Cot[ϑ]
    (r^2 B[r] A′[r]^2 + A[r]^2 (B[r] + r B′[r]) + r A[r] (2 r A′[r] B′[r] + B[r] (A′[r] - r A″[r]))), 0},
 {0, 0, 1/(r^4 A[r]^3 B[r]^4) (4 foft2 r^3 B[r] A′[r]^3 + A[r]^3 (-12 foft2 B[r]^2 +
       (4 foft2 - 2 foft1 r^2) B[r]^3 + foft1 r^2 B[r]^4 + 8 foft2 r B′[r] - 4 foft2 B[r] (-3 + r B′[r])) +
    4 foft2 r^2 A[r] A′[r] (-B[r]^2 A′[r] + r A′[r] B′[r] + B[r] (3 A′[r] - r A″[r])) -
    2 r A[r]^2 (foft1 r^2 B[r]^3 A′[r] - 8 foft2 r A′[r] B′[r] +
       2 foft2 B[r] (A′[r] (-4 + r B′[r]) + 2 r A″[r]) + B[r]^2 (4 foft2 A′[r] - 2 foft2 r A″[r]))),
  0}, {0, 0, 0, 1/(r^4 A[r]^3 B[r]^4) (4 foft2 r^3 B[r] A′[r]^3 + A[r]^3 (-12 foft2 B[r]^2 +
       (4 foft2 - 2 foft1 r^2) B[r]^3 + foft1 r^2 B[r]^4 + 8 foft2 r B′[r] - 4 foft2 B[r] (-3 + r B′[r])) +
    4 foft2 r^2 A[r] A′[r] (-B[r]^2 A′[r] + r A′[r] B′[r] + B[r] (3 A′[r] - r A″[r])) -
    2 r A[r]^2 (foft1 r^2 B[r]^3 A′[r] - 8 foft2 r A′[r] B′[r] +
       2 foft2 B[r] (A′[r] (-4 + r B′[r]) + 2 r A″[r]) + B[r]^2 (4 foft2 A′[r] - 2 foft2 r A″[r])))}}
```